
 Droits et restrictions d’usage :
Les informations contenues dans ce document ont été élaborées par l’IFEN, à partir de la base géographique
Bd_Carthage,  pour répondre à ses propres besoins d’étude et d’analyse à l’échelle nationale. Elles n’ont pas fait l’objet
de validations externes. Leur utilisation, les  méthodes et outils présentés dans dossier devront donc mentionner
explicitement la source (cf ci-dessous - bas de page) et indiquer  les « précautions d’usage » définies dans ce document.
La base géographique Bd_Carthage utilisée ici est la propriété du MATE (Ministère de l’Aménagement du Territoire et
de l’Environnement) et de l’IGN (Institut Géographique National). Elle est protégée par copyright.
Sources :   MATE et agences de l’eau -BD Carthage   (1997)

1. Objectif
Ce projet de détermination de la filiation des zones hydrographiques a été réalisé par l’Ifen dès 1998 dans le
but de répondre dans un premier temps à la nécessité de déterminer les surfaces drainées aux stations de
qualité des eaux. En effet, cette information n’est actuellement (début 1998) disponible dans la base des
stations de qualité fournie par l’OIEau que pour les zones hydrographiques des agences Loire-Bretagne et
Rhône-Méditerranée-Corse. D’où la nécessité de généraliser à tout le territoire.

De façon plus générale, la constitution de l’arborescence des zones hydrographiques permet de connaître
pour chaque zone celles qui sont à son amont par l’établissement de filiations Aval  Amont. On peut ainsi
déterminer la surface drainée en amont mais également estimer toute autre « donnée connue » sur les zones
hydrographiques correspondantes. Ainsi toute donnée administrative (population par communes, données du
Recensement Agricole par cantons, …) peut être estimée par un traitement géographique préalable
d’intersection avec ces zones amont. Les techniques de clé d’affectation sont multiples : pro rata surfacique,
proportionnalité en termes de types de couvertures des terres issus de CORINE land cover,…

2. Méthodologie

2.1. Méthode automatique

Dans la base géographique des zones hydrographiques de Bd-Carthage v2.4, il y a deux champs :
« Drainppal » -code hydrographique générique du drain principal de cette zone,
« Pkmexut » -point kilométrique (Pk traduisant la distance à l’embouchure) à l’exutoire de cette zone.

Rappel : La codification hydrographique d’un cours d’eau est une clé de 8 caractères (cf. Dictionnaire de données
SANDRE du référentiel hydrographique) : ZzzzCccM : Zone hydrographique - Cours d’eau - Milieu.

L’exploitation dans un premier temps du code du drain principal, permet de regrouper toutes les zones
hydrographiques traversées par un même drain, le Pk des zones permettant de classer les zones d’un même
groupe dans le sens amont-aval.

La seconde étape consiste à résoudre le problème des confluences. Chaque zone ne peut posséder qu’un seul
drain principal telle qu’est conçue Bd_Carthage. Or, afin d’effectuer les raccordements entre cours d’eau
principaux et affluents (eux-mêmes drains principaux pour les zones hydrographiques qu’ils drainent), il faut
connaître pour les zones où existe une confluence, le code de tous ces différents drains. La résolution de cet
aspect peut être effectué avec un SIG (ArcInfo ou MapInfo) en déterminant sur quelles zones les différents
drains principaux « s’intersectent ».

Cette première méthode d’approche qui présentant l’avantage d’être automatisable, a vite été abandonnée du
fait que les champs drain principal et Pk exutoire ne sont renseignés dans la base Bd_Carthage que pour les
zones hydrographiques des agences Adour-Garonne et Rhin-Meuse (où d’ailleurs, le code n’est pas toujours
correctement renseigné. Principalement le code cours d’eau).

Il a donc fallu développer une seconde méthode, plus longue, reposant sur un traitement manuel.

2.2. Méthode spécifique

Elle a consisté à repérer visuellement sous MapInfo les relations entre zones hydrographiques. Ce repérage a
été réalisé après avoir superposé les couches des zones hydrographiques de la Bd_Carthage et des cours
d’eau issus des cartes linéaires de qualité des cours d’eau publiées par le RNDE.

3. Table des relations Aval →Amont
A partir du repérage des liaisons entre zones hydrographiques, une table sous ACCESS 97 a été construite (by
C. Germain / IFEN en 1998 – elle est également au format *.DBF). Elle est constituée d’un premier champ contenant le code

des zones possédant au moins une zone hydrographique amont, dont les codes sont regroupés dans un second
champ. Cette structure implique qu’une zone est dupliquée dans le premier champ autant de fois qu’elle
possède des zones amont (cf. Tab-1).

Table Niveau Carte des zones hydrographiques (visualisation des codes)

Aval Amont
Y258 Y257 1
Y257 Y255 2
Y257 Y256 2
Y255 Y253 3
Y255 Y254 3
Y253 Y251 43
Y253 Y252 4
Y251 Y250 5 n

Tab- 1
5816 lignes

Matérialisation de l’arborescence (niveau 1 à 3 sur cet exemple)

Fig- 1 : Production du fichier arborescence des zones hydrographiques

NB. Cette étape « manuelle » est relativement simple à réaliser (< semaine y a été consacré par l’IFEN) mais demande
beaucoup d’attention. Il faut en effet créer la table en même temps que l’on «travaille » surs les cartes et « lever » les
ambiguïtés d’affectation de certains amonts.
La table obtenue est «un classique » de toute relation type : père - fils ,  entreprise - établissements, aval - amonts.

4. Elaboration de l’arborescence
La table initiale des Aval Amont (cf. Table  Tab-1) se prête facilement à des traitements automatisés de type
résolution de réseau. Dans le cas présent c’est un programme SAS qui a été développé pour constituer
l’arborescence des zones hydrographiques (cf. Table  Tab- 2) et produire des statistiques correspondantes.  Il
est donc « facile et rapide », suite à des  modifications ou corrections dans cette table initiale (expertise par
exemple, mise à jour, )  de réitérer  l’ensemble du processus arborescent  (Le programme SAS annexé s’exécute en
moins d’une mn de traitement sur un Pentium 500).

4.1. Typologie des zones (aval → amont)
A partir de la table ci-dessus (cf. Tab-1) on détermine pour
chaque ZH ses amonts et Avals « éventuels ». Ceci permet
de créer une nouvelle table (cf. Tab- 2) qui contient pour
chaque Zone hydrographique :

ZH : Le code de la Zone Hydrographique
Av_Am : Typologie « aval amont »

00 Isolé (pas aval - pas amont)
01 Pas d'aval - Au moins un Amont
10 Au moins un Aval - Pas d'amont
11 Au moins un Aval - Au moins un Amont

Tab- 2 : Typologie et clés d’arborescence
CLE : Clé de l’arborescence des zones Amont
Long : Longueur totale de la clé (y compris les 5 caractères de la Racine)
Niveau : Indicatif du niveau atteint par rapport à la Racine

ZH Av_Am CLE LONG Niveau
Y258 01 Y258- 5 0
Y257 11 Y258-1 6 1
Y255 11 Y258-11 7 2
Y253 11 Y258-111 8 3
Y251 11 Y258-1111 9 4
Y250 10 Y258-11111 10 5
Y252 10 Y258-1112 9 4
Y254 10 Y258-112 8 3
Y256 10 Y258-12 7 2

Y253O401

Y234Y254

Y2

Y236

Y241
O433

Y257

Y255

Y256

Y242

Y160

Y243

Y258

Y161

4
Y

Y145

La constitution de la typologie
des zones hydrographiques est
l’étape la plus importante de ce
traitement puisqu’elle permet de
déterminer les racines (« les têtes
de réseau » du réseau
hydrographique, c’est à dire les
zones qui n’ont pas d’aval mais
au moins un amont. La
détermination des niveaux et
rangs d’arborescence qui suivent
« part » des ces « racines » ; ce
sont donc elles qui conditionnent
tout « le reste ».

L’arborescence des zones hydrographiques part des racines (typologie 01 = zones le plus en aval) pour conduire à
l’autre extrémité  aux zones « sources » (typologie 10).  Entre les deux, les zones 11 sont des nœuds intermédiaires. Les
zones 00, sans aucune liaison avec quelque zone que ce soit, sont des zones isolées.

Fig- 2 : Typologie des zones hydrographiques

4.2. Génération des niveaux d’arborescence
On part des « ZH Racine » (sélection des typologies 01) qui constituent le niveau 1 de l’arborescence (ici
Y258).  Chacune de ces racines a un ou plusieurs amonts « directs » (Rang : 1 à 5 dans le cas des ZH
étudiées ici). Elles constitueront le préfixe des clés d’arborescence (cf : Y258- dans Tab- 2 ).

Partant du niveau 1, on réitère le processus pour chercher ses Amonts qui constitueront ainsi le niveau 2. Le
processus boucle ainsi jusqu’aux nœuds terminaux (plus d’amont) qui constituent les feuilles.
Chaque zone hydrographique se voit ainsi attribuer une Racine et un niveau.

4.3. Génération des clés d’arborescence
Au cours de chacune des itérations (donc à chaque niveau) on génère une clé unique d’arborescence pour la
zone hydrographique en cours de traitement (c’est un des amonts ou une feuille).
La clé est constituée pour chaque niveau atteint n par : le code racine suivi du n° de rang  r :

Niveau
Racine 1 2 3 4 5
Y258- Y258-1 Y258-11  Y258-111 Y258-1111 Y258-11111

Y258 Y257 Y255  Y253 Y251 Y250
Y258-1112

Rangs Y252
 Y258-112

Y254
Y258-12

Y256

Exemple pour la zone Y252
de clé  :  Y258-1112

Rang 1 au niveau 1 ,
Rang 1 au niveau 2 ,
Rang 1 au niveau 3 ,
Rang 2 au niveau 4

R1 N1 R1 N2
R2 N2

R1 N3
R2 N3

R1 N4
R2 N4

R1 N5

Nb : Le niveau maximal atteint dans Bd_Carthage est 87 pour la zone M842 (ceci correspond  à 1018 zones en amont)

5. Exploitation

Par l’exploitation de la clé d’arborescence, il
est possible de connaître pour n’importe
quelle zone toutes les zones amont en
remontant jusqu’aux sources par une simple
requête SQL.  Il « suffit » de sélectionner
toutes les zones ayant comme clé le préfixe
de celle de la zone étudiée.

Left$(CLE,4)= ’’Y258’’
 Sélection des zones Amont de Y258
soit Y257 Y255 Y256 … Y2250

Left$(CLE,8)= ’’Y258-111’’
 Sélection  des zones Amont de Y253

Fig- 3 : Codes d'arborescence des zones hydrographiques (depuis Y258)
Table MapInfo $Data\MapInfo\Hydro\Zoneh.Arborescence.Tab

6. Limites
L’arborescence ainsi élaborée par l’IFEN vise à répondre d’abord à ses besoins d’élaboration de statistiques
à l’échelle nationale. Elle est « manuelle »  en ce sens qu’elle affecte d’abord à chaque zone un ou n amont à
partir de la visualisation d’éléments géographiques (zones, cours d’eau). Il faut donc pouvoir valider certains
amonts « ambigus » (on l’affecte à telle ou telle zone ?) ou confluences intra zonales (on part à gauche ou à
droite ?). Ces cas tiennent à la fois à la géographie des zones hydrographiques elles même mais aussi à la
connaissance précise « du terrain ».

Les contrôles réalisés en interne par l’IFEN, les traitements programmés et les résultats obtenus, permettent
de considérer que cette arborescence est globalement « correcte » et permet de dresser sans biais notable des
statistiques à l’échelon national.
Il faut donc, avant toute utilisation, bien prendre en compte les limites que l’on peut décliner en 3 ensemble.

 La première limite est liée, comme on l’a décrit, à la méthode elle même qui repose sur une étape manuelle
pouvant introduire, malgré les précautions prises, des incohérences dans l’affectation de certaines zones.

 La deuxième limite est inhérente à la conception même de Bd-Carthage puisque les cours d’eau
transfrontaliers ne sont pas munis d’aires en dehors des frontières. C’est le cas en particulier, pour ne citer que
les plus connus, des cours d’eau : Rhône - Doubs - Bidassoa - Garonne - Semois Franco-Belge.

 La troisième limite tient à la précision maximale possible. Le calcul des surfaces aux points RNB ne peut être
déterminé qu’à la dernière zone près. Dans le cas des points intra zonaux, on a supposé que le code
hydrographique du point de mesure respectait la syntaxe pour l’Agence de l’Eau considérée.

Surfaces drainées en km²

7. Perspectives
Partant de la table des zones qui ont un amont, la méthode pour constituer l’arborescence est totalement
automatisée et peut être appliquée à tout type de réseau bâti sur le même principe (routier, électrique, …).
Ceci offre de nombreuses perspectives d’études sur les zones dont on connaît leur « amont drainé »

 La table des arborescences devrait faciliter la codification des drains principaux, rendant ainsi
automatisable (ou presque) la liaison des zones hydrographiques (cf. § 2-1).

 La constitution de cette arborescence trouve tout son intérêt dans le calcul et le cumul de données
propres à des cours d’eau, à des zones, sous-secteurs, secteurs et régions hydrographiques : flux de
matière d’amont en aval, surfaces drainées aux stations de qualité (cf. fig- 4) …

 Les traitements de toute nature réalisés à partir de la table des arborescences peuvent de façon
directe être associés à la cartographie des zones mises en jeu. Ainsi la sélection de zones sur
critères pré définis (vérifiant telle propriété) permet celle de ses zones amont et donc leur
cartographie (au sens de l’arborescence définie ici).

fig- 4 : Surfaces drainées aux stations de Qualité

8. Annexes

8.1. Quelques chiffres

Typologies
de zones

Signification Nombre  de
zones

Zones Bd_Carthage correspondantes
 (une zone peut être multi polygonale)

00 Isolé (pas aval - pas amont) 268 282 4%
01 Pas d'aval - Au moins un Amont 127 143 2%
10 Au moins un Aval - Pas d'amont 2080 2107 33%
11 Au moins un Aval - Au moins un Amont 3714 3780 60%

∑ 6189 6312 100%

Racines  127  (soit 143 polygones Zh en  Bd_Carthage). La plus grande M842  « draine » 1028 zones

Niveaux  87 au maximum -  atteint pour la zone racine « M842 »

Rangs  C’est le nombre d’amonts qui sont présents à chaque niveau » (1 à 5 dans notre cas)
1 1959 1959
2 1810 3620 dernière colonne = rang * nombre de zones
3 68 204
4 7 28
5 1 5

∑ 3845 5816 On retrouve le nombre de lignes de la table Aval_Amont (ci-dessous)

Nombre
maximum de
Zones amont

Nombre de zones que l’on trouve en amont de la racine qui a l’ascendance la plus grande.
1018 zones  maximum pour la « racine M842 » (une grande partie du Bassin L.B)

Durées Une semaine de travail « manuel » pour constituer la table des « Aval - Amonts»
1 minute CPU sur Pentium 500 pour générer les clés d’arborescence

8.2. Principales informations produites

Toute l’information produite se résume à deux tables.

Table initiale des Aval - Amont Table finale des Arborescences
Aval Amont
Y258 Y257
Y257 Y255
Y257 Y256
Y255 Y253
Y255 Y254
Y253 Y251
Y253 Y252
Y251 Y250

Arbo_Aval_Amont.Dbf 5816 lignes Zoneh_Arborescence.Dbf                   6189 lignes

Réalisée par l’IFEN (1998 ; C. Germain)
« manuellement » à partir des couches :
zones hydrographiques Bd_Carthage -
cours d’eau des cartes linéaires de qualité
des cours d’eau publiées par le RNDE

Réalisée par l’IFEN  en développant les
méthodologies et les modules (programmation
SAS) de traitement correspondants.
Processus entièrement « automatisé »

ZH Av_Am CLE LONG Niveau
Y258 01 Y258- 5 0
Y257 11 Y258-1 6 1
Y255 11 Y258-11 7 2
Y253 11 Y258-111 8 3
Y251 11 Y258-1111 9 4
Y250 10 Y258-11111 10 5
Y252 10 Y258-1112 9 4
Y254 10 Y258-112 8 3
Y256 10 Y258-12 7 2

8.3. Programme SAS pour générer les « arborescences »  hydrographiques
/*=====================================================================================*/
/*= ZONES HYDROGRAPHIQUES (Bd_Carthage)    Traitements    =============================*/
/*-------------------------------------------------------------------------------------*/
/*                                                                                     */
/* Arborescence des zones hydrographiques                                              */
/*                                                                                     */
/* On part de la table "manuelle" qui associe AVAL-AMONT pour chaque ZH                */
/* On part de la table des zones hydrographiques (extrait. de la base géographique)    */
/*                                                                                     */
/* On crée la typologie de zones suivante :                                            */
/*    00   ZH isolée (pas d'AVAL ni d'AMONT)                                           */
/*    01   ZH sans AVAL        - au moins un AMONT                                     */
/*    10   ZH sans AMONT       - au moins un AVAL                                      */
/*    11   ZH au moins un AVAL - au moins un AMONT                                     */
/*                                                                                     */
/* On sélectionne les ZH Racine                 :                                      */
/*    Celles qui ont la typologie 01 (ce sont les têtes de réseau : Pas d'Aval)        */
/*                                                                                     */
/* On crée pour chaque ZH sa clé d'arborescence :                                      */
/*    ABCD-11212...  où   ABCD est la racine                                           */
/*                        11212 ... Rang de la ZH (à chaque Niveau 1... 87)            */
/*   Nb. Le niveau maximal traité est de 87 (la feuille se trouve au niveau 87)        */
/*   Nb. Le rang   maximal traité est de 5  (nombre d'AMONT maxi. pour une ZH)         */
/*                                                                                     */
/*-------------------------------------------------------------------------------------*/
/*Ifen Dstat/Si – Méthodes & Techniques   Arborescence de Zones Hydrographiques    */
/*    M. Lacaze     19 janvier 1998 - 24 Janvier 2001                                  */
/*=====================================================================================*/

/*=1===================================================================================*/
/*= Listes des zones hydrographiques extraites de la base géographique Bd_Carthage    =*/
/*=====================================================================================*/

proc access dbms=dbf;
     create work.NT.access;
     path='T:\Eau\Zones\Liste_ZH.Dbf';
     assign=yes;
     create work.NT.view;
     select all;
run;
Data Lis_ZH (Keep=Aval);
     Length Aval $ 4.;
     Set NT;
     Aval=ZH;
Run;
Proc Summary Data=Lis_ZH Nway;
     Class Aval;
     Output out=Lis_ZH;
run;

Proc sort data=Lis_ZH; by Aval; Run;
Proc print data=Lis_Zh; run;

/*=2===================================================================================*/
/*= Tab0  Table  AVAL AMONT    créée manuellement par C. GERMAIN en 1998              =*/
/*= XTab0 Table  AVAL AMONT    Mais où on a inversé AVAL et AMONT par rapport à TAB0  =*/
/*=====================================================================================*/

proc access dbms=dbf;
     create work.NT.access;
     path='T:\Eau\Zones\Arbo_Aval_Amont.Dbf';
     assign=yes;
     create work.NT.view;
     select all;
run;
Data Tab0  (Keep=Aval Amont);
      Length Aval Amont $ 4.;
     Set NT;
Run;

Data XTab0 (Keep=Aval Amont); /* Meme table où on inverse AVAL et AMONT */
     Retain X_Sav '----';
     Set Tab0;
     X_Sav=Amont;
     Amont=Aval;
     Aval=X_Sav;
Run;

Proc Summary Data=tab0 Nway;

     Class Aval;
     Output out=tab00;
run;

Proc Summary Data=Xtab0 Nway;
     Class Aval;
     Output out=Xtab00;
run;

/*=2.1=================================================================================*/
/*= Typologies 01 et 11                                                               =*/
/*=     AVAL (de TAB0) a forcément un AMONT     --> Av_AM = 01                        =*/
/*=     Si AVAL (de TAB0) est aussi en AMONT (i.e AVAL de XTAB00)   --> Av_AM = 01    =*/
/*=====================================================================================*/

Proc sort data=Tab00; by Aval; run;
Proc sort data=XTab00; by Aval; run;

Data Z_0 (Keep=Aval Av_Am);
     Retain Av_Am '00';
     Merge Tab00 (in=a) Xtab00 (in=b); By Aval;
     Av_Am='01';
     If a then do;
        If b then Av_am='11';
     End;
     If a;
Run;

Proc Summary Data=Z_0;
     Class Av_Am;
     Output out=XY;
run;
Proc print; Run;

/*=2.2=================================================================================*/
/*= Typologies 10                                                                     =*/
/*=     Si AVAL de XTAB00 (i.e AMONT de TAB00) n'est pas dans Z_0 c'est donc qu'il    =*/
/*=     a un AVAL et pas d'AMONT  --> Av_Am = 10                                      =*/
/*=====================================================================================*/

Data Z_00 (Keep=Aval Av_Am);
   Merge Z_0 (in=a) Xtab00 (in=b); By Aval;
   If a then do;
      Output;
   End;
   Else do;
      If b then do;
         Am_n=0;
         Av_Am='10';
         Output;
      End;
   End;
Run;

Proc Summary Data=Z_00;
     Class Av_Am;
     Output out=XY;
run;
Proc print; Run;

/*=2.3=================================================================================*/
/*= Typologies 00                                                                     =*/
/*=     La fusion entre la liste des ZH (LIS_ZH) et la table Z_00 permet de déterminer=*/
/*=     le cas des ZH isolées --> Av_Am = 00                                          =*/
/*=====================================================================================*/
Data ZH (Keep=Aval Av_Am);
   Retain Av_Am;
   Merge Lis_ZH (in=a) Z_00 (in=b); By Aval;
   If a & ^b Then do;
      Av_Am='00';
   End;
   If a;
Run;
Proc Summary Data=ZH;
     Class Av_Am;
     Output out=XY;
run;
Proc print; run;

/*=3===================================================================================*/
/*= Racines   (ZH terminales) = Tête de réseau hydrograohique (a des Amonts, pas Aval)=*/
/*=====================================================================================*/
Proc sort data=Tab0; by Aval; run;
Data Racines;

   Merge Tab0  ZH (in=b); By Aval;
   If b;
   If Av_Am = '01';
Run;
Proc print; run;

Proc summary data=Racines;
   Class Aval;
   Output out=xx;
Run;

Title 'Tetes de zones Hydro.';
Proc print; run;

/*=4===================================================================================*/
/*= Cle  d'arborescence pour chaque ZH (qui a au moins un AMONT)                      =*/
/*=      On Travaille par niveaux en partant de la RACINE                             =*/
/*=      Niveau 1 = AMONT(s) de la Racine - Rangs 1 à n                               =*/
/*=      Niveau 2 = AMONT(s) du Niveau 1  - Rangs 1 à n                               =*/
/*=      ...                                                                          =*/
/*=      Niveau 87= AMONT(s) du Niveau 86 - Rangs 1 à n                               =*/
/*= Nb   Dès le dernier niveau  atteint le processus n'en génère plus de nouveaux     =*/
/*=====================================================================================*/
Data N1 (Keep=Niveau Rang Aval Amont CAval);
     Length Caval $ 110.;
     Retain Rang 0 Niveau 1;
     Set Racines; By Aval;
     Rang=Rang+1;
     Substr(Caval,Niveau,1)=Put(Rang,Z1.);
     Output;
     If Last.aval then do;
        Rang=0;
     End;
Run;

%Let Debut=2;
%Let Fin=87;
%Macro Niveaux;
   %Do Niveau = &Debut %to &Fin;
      Data N&Niveau (Keep=Niveau Rang Aval Amont CAval);
         Retain Premier Dernier Arret I R_Aval 0 Npoint Lect 1;
         Retain Rang 0 Niveau &Niveau;
         Retain Camont '    ';
         Set N%Eval(&Niveau-1);
         Camont=Amont;
         Sav_Av=Aval;
         Output N&Niveau;
         If Niveau = &Niveau - 1;
         Premier=_N_;
         If Amont < Aval then do;
            premier = 1;
         end;
         Rang=0;
         R_Aval=0;
         Do i=Premier to Dernier while (R_Aval=0);
            Set Tab0 point= i Nobs=Dernier;
            Npoint=Npoint+1;
            If Aval=CAmont then do;  /* On a trouvé !!!!                      */
               Rang=Rang+1;
               Niveau=&Niveau;
               Substr(Caval,&Niveau,1)=Put(Rang,Z1.);
               Aval=Sav_Av;
               Output N&Niveau;
             End;
             Else do;
               If Aval > CAmont then R_Aval=-1; /* on ne trouvera jamais      */
            End;
         End;/* DO i = Premier to Dernier ...*/
       Run;
  %End;
%Mend Niveaux;

%Niveaux

Proc sort data=N&Fin; By Amont; Run;
Proc print data=N&Fin ; run;

/*=4.1=================================================================================*/
/*= Cle  Génération de la clé et de sa longueur                                       =*/
/*=====================================================================================*/
Data Arbo (Keep= Aval ZH CLE LONG);
  Length Cle $ 100;
  Retain ZH '    ' Long 0;
  Set N&Fin;
  ZH=AMONT;

  CLE=Aval|| '-' || Substr(CAval,1,Niveau);
  Long=Niveau+5;
  Aval=Amont;
Run;

Proc print; run;

/*=4.1=================================================================================*/
/*= Intégration de toutes les ZH (y.c celles qui sont hors arborescence)              =*/
/*=====================================================================================*/
Proc sort data=Arbo; by aval; run;
Data ZH1 (Keep=Av_Am ZH CLE LONG);
   Merge ZH (in=a) Arbo (in=b); By Aval;
   If a & ^b then do;
      Zh=Aval;
      CLE=Aval;
      Long=4;
   End;
   If a;
Run;

Proc Summary Data=ZH1;
     Class Av_Am;
     Output out=CF;
run;
Proc print; run;

%Let DBF    = T:\Eau\Zones\Zoneh_Arborescence.Dbf.DBF;
Proc dbload dbms= dbf data=ZH1;
   Errlimit=0;
   Limit=0;
   Path="&DBF";
   Version= 4;
   Load;
Run;


